

دانشگاه تهران - دانشکده فنی
گروه مهندسی برق و کامپیوتر

خودآموز زبان توصیف سخت‌افزاری
Verilog

تهیه کننده : سعید صغری

اسفند ۷۹




۱ مفاهیم اولیه

۱-۱ ماجول

ماجول بلوک پایه Verilog است. یک ماجول می‌تواند یک عنصر یا مجموعه‌ای از بلوکهای سطح پایین‌تر باشد. بنابراین ماجول عملکرد مورد نظر را برای بلوکهای سطح بالاتر فراهم می‌کند، اما پیاده‌سازی داخل آنرا پنهان می‌کند. شکل ۱-۱ نحوه تعریف ماجول را با یک مثال بیان می‌کند. در زبان Verilog می‌توانیم یک ماجول را در چهار سطح مختلف تجرید بیان کنیم:


- **سطح گیت**: در این سطح ماجول بصورت گیت‌های منطقی و اتصالات بین آنها بیان می‌شود.
- **سطح جریان داده (Dataflow)**: در این سطح ماجول بوسیله مشخص کردن نحوه جریان اطلاعات بین رجیسترها و نوع پردازشی که روی آنها صورت می‌گیرد، بیان می‌شود.
- **سطح رفتاری (Behavioral)**: در این سطح ماجول برحسب الگوریتم طراحی شود، بدون اینکه جزئیات طراحی پیاده‌سازی سخت‌افزاری در نظر گرفته شود.
- **سطح سوئیچ**: در این سطح ماجول بصورت سوئیچها و اتصالات بین آنها بیان می‌شود.



شکل ۱-۱- نحوه تعریف ماجول


۲-۱ نمونه

یک ماجول الگویی از یک عنصر واقعی می‌سازد، هنگامیکه از این ماجول استفاده می‌شود، Verilog یک نمونه از این الگو می‌سازد. هر عنصر دارای نام، متغیرها و پارامترهای خاص خود است. پروسه ایجاد یک نمونه از الگوی یک ماجول را اصطلاحاً **Instantiation** یا نمونه‌سازی و این عنصر را **Instance** یا نمونه می‌نامند. بعنوان مثال در شکل ۲-۱ نحوه ساخت یک نمونه از روی ماجول دیکودر را می‌بینیم.



شکل ۱-۲- نحوه ساخت نمونه از روی ماجول دیکودر

- در شکل زیر یک مقایسه‌گر بیتی نشان داده شده است. یک ماجول بنام **Comparator** تعریف کنید و سپس یک نمونه بنام **Comp0** از روی آن بسازید. دقت کنید هدف فقط تعریف پورت‌های مقایسه‌گر و نمونه‌سازی از روی آن است.



مقایسه‌گر بیتی

پرسش ۱

۲ قراردادهای نحوی

قراردادهای نحوی که بوسیله Verilog استفاده می‌شود، بسیار شبیه زبان برنامه‌نویسی C است. هر توکن می‌تواند توضیح، جداکننده، عدد، رشته، شناسه و یا کلمه کلیدی باشد. Verilog یک زبان حساس به متن است و تمام کلمات کلیدی آن با حروف کوچک نوشته می‌شوند.

۱-۲ حروف فاصله

حروف فاصله در Verilog عبارتند از: فاصله (b)، tab (t) و خط‌جدید (n). این حروف توسط Verilog نادیده گرفته می‌شوند، مگر اینکه بعنوان جداکننده توکنها استفاده شوند و یا در یک رشته استفاده شوند.



۲-۲ توضیحات

توضیحات برای خواناتر کردن طرح بکار می‌روند و به دو صورت یک و چندخطی استفاده می‌شود. توضیحات نمی‌توانند بصورت تودرتو استفاده شوند.

```
a = b && c; // This is a single line comment
/* This is a multiple line
   comment */
```

۳-۲ اپراتورها

اپراتورها به سه دسته یگانی، دوتایی و سه‌تایی تقسیم می‌شوند و به ترتیب دارای یک، دو و سه اپرند هستند.

۴-۲ مشخصات اعداد

در Verilog اعداد به دو صورت زیر نوشته می‌شوند :

- **عدد اندازه دار :** در اینحالت Verilog را مقید می‌کنیم یک عدد را با همان اندازه موردنظر ما بکار ببرد. شکل عدد اندازه‌دار بصورت $\langle \text{radix} \rangle \langle \text{number} \rangle \langle \text{size} \rangle$ است. $\langle \text{size} \rangle$ به دسیمال نوشته شده و تعداد بیت‌های عدد را مشخص می‌کند. $\langle \text{radix} \rangle$ مبنای عدد را مشخص می‌کند، d یا D برای مبنای ده، b یا B برای مبنای دو، o یا O برای مبنای هشت و h یا H برای مبنای شانزده بکار می‌رود. حالت پیش فرض مبنای ده است. برای خواناتر شدن اعداد می‌توان از "_" استفاده نمود.

```
8'b1010_1110 // This is a 8 bit binary number
12'h66        // This is a 12 bit hex number
16'D255       // This is a 12 bit decimal number
-4'd13        // This is a 4 bit negative decimal number, that stored
               // as 2's complement of 13
```

- **عدد بدون اندازه :** شکل کلی عدد بدون اندازه بصورت $\langle \text{radix} \rangle \langle \text{number} \rangle$ است. در این حالت طول عدد به نوع پیاده‌سازی بستگی دارد ولی حداقل ۳۲ بیت است.

```
1234          // This is a 32 bit decimal number
h'62          // This is a 32 bit hex number
o'255         // This is a 32 bit octal number
```



- اعداد منفی با قرار دادن یک علامت منفی ("−") قبل از `<size>` بدست می‌آیند.
- کرکتر ؟ و X معادلا بکار می‌روند.
- در Verilog دو نماد برای مقادیر HiZ و نامعلوم داریم که به ترتیب عبارتند از X, Z. این مقادیر در شبیه‌سازی مدارها بسیار مورد استفاده قرار می‌گیرند. در مقداردهی به یک عدد در مبنای شانزده، هشت و دو مقادیر X, Z به ترتیب طولی برابر ۴ و ۳ و ۱ بیت دارند.

```
12'h13x      // This is a 12 bit hex number, 4 LSBs unknown
4'b10??      // This is a 4 bit binary number equal to 4'b10xx
```



- مشخص کنید که کدام یک از اعداد زیر معتبر هستند.
- 659 'h83F 4AF 12'hx 5'D3 8'd-6 27_13_45

۲-۵ رشته

مجموعه‌ای است از کرکترها که بوسیله " " محصور شده‌اند.

۲-۶ شناسه و کلمه کلیدی

کلمات کلیدی شناسه‌هایی هستند که از پیش برای تعریف ساختار زبان استفاده شده‌اند. کلمات کلیدی با حروف کوچک نوشته می‌شوند. شناسه‌ها نامهایی هستند که ما به عناصر نسبت می‌دهیم تا بوسیله آن به آنها رجوع کنیم. شناسه می‌تواند از کرکترهای حرفی، _ و \$ تشکیل شود و حتما باید با یک کرکتر حرفی شروع شود. شناسه‌ها حساس به متن هستند.

۲-۷ انواع داده‌ها

۲-۷-۱ مجموعه مقادیر

Verilog برای مدلسازی عملکرد سخت‌افزارهای واقعی از ۴ مقدار و ۸ سطح قدرت استفاده می‌کند. مجموعه مقادیر در جدول ۱-۱ و سطوح قدرت در جدول ۱-۲ آمده است.



Value Level	Condition in Hardware Circuits
0	Logic Zero, False Condition
1	Logic One, True Condition
x	Unknown value
z	High Impedance, Floating State

جدول ۱-۱- مجموعه مقادیر

Strength Level	Type	Degree
supply	Driving	strongest
strong	Driving	
pull	Driving	
large	Storage	
weak	Driving	
medium	Storage	
small	Storage	
highz	High Impedance	Weakest

جدول ۲-۱- سطوح قدرت

Net ۲-۷-۲

برای برقرار کردن ارتباط بین اجزاء سخت‌افزاری بکار می‌رود. درست مانند مدارهای واقعی، net دارای مقداری است که بوسیله خروجی عنصر متصل به آن روی آن درایو می‌شود. net در Verilog توسط کلمه کلیدی wire تعریف می‌شود و مقدار پیش فرض آن z است.

```
wire a; // Declare net a
wire b=1'b0 // Net b is fixed to logic value 0 at declaration
```

- نحوه برخورد با z در ورودیهای گیتها و عبارات دقیقا مانند x است، فقط ترانزیستورهای MOS حالت z را از خود عبور می‌دهند.
- سطوح قدرت برای حل تصادم بین دو درایور مختلف در نظر گرفته می‌شود.
- اگر دو سیگنال نامساوی با سطوح قدرت مختلف یک سیم را درایو کنند، سیگنال قویتر برنده می‌شود.
- اگر دو سیگنال نامساوی با سطوح قدرت مساوی یک سیم را درایو کنند، حاصل نامعلوم می‌شود.





Register ۳-۷-۲

برای ذخیره اطلاعات بکار می‌رود. رجیستر تا وقتی مقدار جدیدی روی آن نوشته نشده مقدار خود را نگاه می‌دارد. برخلاف `net`، `register` به درایور نیاز ندارد. `register` در Verilog توسط کلمه کلیدی `reg` تعریف می‌شود و مقدار پیش‌فرض آن `X` است.

```
reg reset; // Declare a variable that can be hold its value
```

Vector ۴-۷-۲

انواع داده‌ای `reg` و `wire` می‌توانند بصورت بردار تعریف شوند. شکل کلی تعریف بردار بصورت زیر است:

<vector_type> [MSB : LSB] <vector_name>

```
wire [31:0] BusA; // Declare a bus that has 32 bit width  
reg [0:40] Vir_Add; // virtual address 41 bits wide
```

۵-۷-۲ انواع داده‌ای صحیح، حقیقی و زمان

- **صحیح**: یک نوع داده رجیستر همه منظوره است که برای پردازش مقادیر صحیح استفاده می‌شود. تفاوت این نوع با نوع `reg` در اینست که در نوع `reg` داده‌ها بصورت بدون علامت در نظر گرفته می‌شوند، ولی در نوع صحیح داده‌ها بصورت علامتدار در نظر گرفته می‌شوند. نوع داده‌ای صحیح توسط کلمه کلیدی `integer` تعریف می‌شود و طول آن بستگی به پیاده‌سازی دارد ولی حداقل ۳۲ بیت در نظر گرفته می‌شود.

```
integer counter; // General purpose variable used as a counter
```

- **حقیقی**: یک نوع داده رجیستر همه منظوره است که برای پردازش مقادیر صحیح استفاده می‌شود. نوع داده‌ای صحیح توسط کلمه کلیدی `integer` تعریف می‌شود.

```
real delta; // Declare a real variable called delta
```

- **زمان**: در Verilog یک نوع داده‌ای خاص برای ذخیره کردن زمان شبیه‌سازی استفاده می‌شود. نوع داده‌ای زمان بوسیله کلمه کلیدی `time` تعریف می‌شود. طول متغیر از نوع زمان به پیاده‌سازی بستگی دارد ولی حداقل ۶۴ بیت است.

```
time sim_time; // Define a time variable sim_time
```



۸-۲ آرایه

در Verilog می‌توان آرایه‌ای از نوعهای داده‌ای `reg`، `integer`، `time` و یا آرایه‌ای از بردارهایی از این نوعها تعریف کرد. شکل کلی تعریف آرایه عبارتست از:

```
<array_type> <array_name> [ #first_element : #last_element ]
```

```
integer count[0:7]; // An array of 8 integer
```

- در Verilog آرایه‌های چندبعدی نداریم.
- تفاوت آرایه و بردار در اینستکه آرایه از چند عنصر تشکیل شده است که هر یک از آنها می‌توانند یک یا چند بیت طول داشته باشند، درحالیکه بردار یک عنصر است که طولی برابر n دارد.



۹-۲ حافظه

در Verilog حافظه را بصورت آرایه‌ای از رجیسترها تعریف می‌کنیم.

```
reg membit[0:1023]; // 1K x 1bit memory  
reg [7:0] membyte[0:1023]; // 1K x 8bit memory
```

۱۰-۲ پارامتر

می‌توان در یک ماجول اعداد ثابتی را بصورت پارامتر تعریف نمود و از آنها استفاده نمود، این امر توسط کلمه کلیدی `parameter` انجام می‌شود.

```
parameter port_id = 5; // Define a constant port_id
```

۱۱-۲ رشته

رشته‌ها می‌توانند در یک `reg` ذخیره شوند. طول متغیر `reg` باید به اندازه کافی بزرگ باشد تا بتواند رشته را نگاه دارد. هر کرکتر در ۸ بیت ذخیره می‌شود.

```
reg [8*18:1] s_val; // Define a variable with 18 bytes
```

۱۲-۲ task های سیستم

در Verilog تعدادی `task` به منظور نمایش اطلاعات روی صفحه، مانیتورکردن مقادیر، خاتمه شبیه‌سازی و... فراهم شده است. کلیه `task` ها بصورت `<task_name>` بکار می‌روند.



- **نمایش اطلاعات**: نمایش اطلاعات توسط $\$display$ انجام می‌شود و دارای شکل کلی $\$display(\text{format}, p1, p2, \dots, pn)$; که در آن pi ها می‌توانند نام متغیر، نام سیگنال و یا رشته باشند. توسط format می‌توان قالب نمایش اطلاعات را بصورت دلخواه تعیین نمود. برای این منظور یک سری قالب از پیش تعریف شده است که لیست آنها در جدول ۱-۳ آمده است.

Format	Display
%d or %D	Display variable in decimal
%b or %B	Display variable in binary
%s or %S	Display string
%h or %H	Display variable in hex
%c or %C	Display ASCII character
%v or %V	Display strength
%o or %O	Display variable in octal
%t or %T	Display in current time format
%e or %E	Display real number in scientific format
%f or %F	Display real number in decimal format
%g or %G	Display real number in scientific or decimal format, whichever is shorter

جدول ۱-۳- لیست مشخصات قالبها

```
$display("At time %t virtual address is %h", $time, vir_Adr );
```

- **مانیتور کردن اطلاعات**: مانیتور کردن اطلاعات توسط $\$monitor$ انجام می‌شود و دارای شکل کلی $\$monitor(\text{format}, p1, p2, \dots, pn)$; که در آن pi ها می‌توانند نام متغیر، نام سیگنال و یا رشته باشند. تفاوت نمایش اطلاعات با مانیتور کردن اطلاعات در اینستکه، $\$display$ با هر بار فراخوانی یک مرتبه مقادیر پارامترهایش را نشان می‌دهد ولی $\$monitor$ بطور دائم مقادیر پارامترهایش را مانیتور می‌کند و به محض تغییر یکی از پارامترهای آن، مقادیر کلیه پارامترهایش را نشان می‌دهد. باید توجه داشت که در هر لحظه فقط یک $\$monitor$ می‌تواند فعال باشد، بنابراین اگر چندین دستور $\$monitor$ داشته باشیم، فقط آخرین دستور $\$monitor$ فعال است. توسط دو task به نامهای $\$monitoron$ و $\$monitoroff$ می‌توان عملیات مانیتور کردن اطلاعات را به ترتیب فعال و غیرفعال نمود.
- **توقف و خاتمه شبیه‌سازی**: توسط $\$stop$ می‌توان عملیات شبیه‌سازی را متوقف نمود و آنرا در مد interactive قرار داد. این مد برای عیب‌یابی بکار می‌رود. توسط $\$finish$ می‌توان عملیات شبیه‌سازی را خاتمه داد.

۱۳-۲ راهنمای کامپایلر

در Verilog راهنمای کامپایلر دارای شکل کلی $\langle \text{keyword} \rangle$ است. دو نوع راهنمای کامپایلر

که مورد استفاده بیشتری دارند، عبارتند از:



- `define` که برای تعریف ماکرو بکار می‌رود.
- `include` که برای الحاق یک فایل Verilog به فایل جاری بکار می‌رود.

```
`define WORD_SIZE 32          // Used as `WORD_SIZE in the code
`include "header.v"          // Include the file header.v
```

۳ ماجول

قبلاً نحوه تعریف و نمونه‌سازی ماجول را دیدیم، در اینجا بصورت دقیقتر اجزا ماجول را مورد بررسی قرار می‌دهیم. در شکل ۱-۳ اجزا ماجول مشخص شده است. قسمتهایی که بصورت زیرخط‌دار نوشته شده‌اند، در تعریف ماجول ضروری و سایر قسمتها اختیاری است. قسمتهایی که بدنه ماجول را تشکیل می‌دهند، می‌توانند با هر ترتیبی در تعریف ماجول استفاده شوند.

```
module module_name ( port list );
    port declarations (if ports present)
    parameters (optional)
```

Declaration of wires, regs and other variables

Data flow statement (assign)

Instantiation of lower level module

always and initial blocks, all behavioral statements go in these blocks

tasks and functions

```
endmodule
```


شکل ۱-۳- اجزاء تشکیل دهنده ماجول

۴ پورت

پورتهای یک ماجول، واسط ارتباطی ماجول با جهان خارج است.

۴-۱ لیست پورتهای

به هنگام تعریف ماجول دیدیم که لیست پورتهای (در صورت وجود) در جلوی نام ماجول معرفی می‌شوند. بعنوان مثال در شکل ۱-۴ یک جمع‌کننده ۴ بیتی و نحوه تعریف پورتهای آن نشان داده شده است.



شکل ۴-۱- جمع‌کننده ۴ بیتی و نحوه تعریف پورت‌های آن

۴-۲ تعریف پورت‌ها

تمام پورت‌های یک ماچول باید تعریف شوند. این تعریف طبق جدول ۴-۱ صورت می‌گیرد.

Verilog Keyword	Type of port
input	Input Prt
output	Output Prt
inout	Bidirectional Prt

جدول ۴-۱- تعریف پورت‌ها

- کلیه پورت‌ها بطور ضمنی بصورت **wire** تعریف می‌شوند، مگر اینکه بخواهیم یک خروجی مقدارش را نگاه‌دارد که در اینصورت آنرا از نوع **reg** تعریف می‌کنیم.



بعنوان مثال در مورد جمع‌کننده شکل ۴-۱ تعریف پورت‌ها بصورت زیر است.

```
module fulladd4( sum, cout, a, b, cin );
  output [3:0] sum;
  output cout;

  input [3:0] a, b;
  input cin;

  < Module Body >

endmodule
```

۴-۳ قوانین اتصال پورت‌ها

به هنگام اتصال پورت‌های یک ماچول به جهان خارج باید به نکاتی توجه داشت :

- پورت‌های ورودی ماچول باید از نوع **net** باشند و این پورت‌ها می‌توانند به متغیرهایی از نوع **reg** و یا **net** درجهان خارج متصل شوند.
- پورت‌های خروجی ماچول می‌توانند از نوع **reg** و یا **net** باشند و این پورت‌ها باید به متغیرهایی از نوع **net** درجهان خارج متصل شوند.



- پورت‌های دوسویهٔ ماجول باید از نوع **net** باشند و این پورت‌ها می‌توانند به متغیرهایی از نوع **reg** و یا **net** درجهان خارج متصل شوند.
 - پورت‌های ماجول و متغیرهای خارجی متصل به آنها باید از نظر طول منطبق باشند.
- در Verilog به دو صورت می‌توان ارتباط پورت‌ها را با جهان خارج برقرار نمود:
- ۱- **اتصال ترتیبی**: در این روش به هنگام نمونه‌سازی از یک ماجول، متغیرهای متصل به پورت‌ها را دقیقاً به همان ترتیبی که در تعریف ماجول آمده‌اند، بیاوریم. در این روش اگر بخواهیم یک پورت خروجی به جایی متصل نباشد کفیسست جای آن را در لیست خالی بگذاریم.

۲- **اتصال از طریق نام**: در این روش برای اتصال هر پورت از قالب زیر استفاده می‌کنیم:

`.port_name(external_signal_name)`


از این روش وقتی استفاده می‌کنیم که تعداد پورت‌ها زیاد باشد و بخاطر سپردن ترتیب آنها دشوار

باشد. در این روش اگر بخواهیم یک پورت خروجی به جایی متصل نباشد کفیسست نام آن را نیاوریم.

```
//-- First Method -----
fulladd4 fa0( s, co, x, y, ci );
fulladd4 fa0( s, , x, y, ci );

//-- Second Method -----
fulladd4 fa0( .sum(s), .cout(co), .cin(ci), .a(x), .b(y) );
fulladd4 fa0( .sum(s), .cin(ci), .a(x), .b(y) );
```

در شکل ۵-۱ قوانین اتصال پورت‌ها نمایش داده شده است.



شکل ۵-۱- قوانین اتصال پورت‌ها

۵ مدل‌سازی در سطح گیت

در مدل‌سازی در سطح گیت مدار را بصورت مجموعه‌ای از گیت‌های پایه که به یکدیگر متصل شده‌اند بیان می‌کنیم. برای این منظور باید انواع گیت‌های پایه‌ای که در Verilog وجود دارند را معرفی کنیم.


۱-۵ گیت‌های پایه

۱-۱-۵ and/or گیت‌های

جدول صحت گیت‌های پایه **and/or** در شکل ۶-۱ آمده است.



- با اتصال چهار مقایسه‌گر یک بیتی مطابق شکل زیر، یک ماجول مقایسه‌گر چهاربیتی بنام Comparator4 بسازید. فرض کنید که ماجول مقایسه‌گر یک بیتی بنام Comparator موجود است. سیگنالهای داخلی مورد نیاز را از im0 تا im8 نامگذاری کنید.



پرسش ۴

and	0	1	x	z
0	0	0	0	0
1	0	1	x	x
x	0	x	x	x
z	0	x	x	x

or	0	1	x	z
0	0	1	x	x
1	1	1	1	1
x	x	x	1	x
z	x	1	x	x

xor	0	1	x	z
0	0	1	x	x
1	1	0	x	x
x	x	x	x	x
z	x	x	x	x

nand	0	1	x	z
0	1	1	1	1
1	1	0	x	x
x	1	x	x	x
z	1	x	x	x

nor	0	1	x	z
0	1	0	x	x
1	0	0	0	0
x	x	0	x	x
z	x	0	x	x

xnor	0	1	x	z
0	1	0	x	x
1	0	1	x	x
x	x	x	x	x
z	x	x	x	x

شکل ۱-۶- جدول صحت گیت‌های پایه and/or

```

and a1( out, in1, in2 );           // and gate with two inputs
or ( z, i1, i2 );                 // or gate with two inputs
xor ( z1, i1, i2, i3, i4 );       // xor gate with four inputs
nand nd( o1, i1, i2, i3 );        // nand gate with three inputs
nor ( out, a, b, c );             // nor gate with three inputs
xnor ( y, a, b, c, d );          // xnor gate with four inputs

```

- پورت اول این گیتها، پورت خروجی و بقیه پورتها ورودی هستند. به این ترتیب می‌توان گیت‌هایی با تعداد ورودیهای دلخواه داشت.
- به هنگام نمونه‌سازی (استفاده) از این گیتها، می‌توان نام نمونه را ذکر نکرد (این امر در مورد تمام المانهای از پیش ساخته شده Verilog صادق است).

تذکره

**۲-۱-۵ گیت‌های buf/not**

جدول صحت گیت‌های پایه buf/not در شکل ۷-۱ آمده است.

buf		ctrl					ctrl				
buf	out	bufif0	0	1	x	z	bufif1	0	1	x	z
0	0	0	0	z	L	L	0	z	0	L	L
1	1	1	1	z	H	H	1	z	1	H	H
x	x	x	x	z	x	x	x	z	x	x	x
z	x	z	x	z	x	x	z	z	x	x	x

not		ctrl					ctrl				
not	out	notif0	0	1	x	z	notif1	0	1	x	z
0	1	0	1	z	H	H	0	z	1	H	H
1	0	1	0	z	L	L	1	z	0	L	L
x	x	x	x	z	x	x	x	z	x	x	x
z	x	z	x	z	x	x	z	z	x	x	x

شکل ۷-۱- جدول صحت گیت‌های پایه buf/not

- پورت آخر گیت‌های buf, not, پورت ورودی و بقیه پورت‌ها خروجی هستند. به این ترتیب می‌توان گیت‌هایی با تعداد خروجی‌های دلخواه داشت.
- در مورد گیت‌های دیگر پورت اول خروجی، پورت بعدی ورودی و پورت سوم کنترل است.
- L به معنای 0 یا z و H به معنای 1 یا z می‌باشد.

نکته

```
buf b1 ( o1, o2, i1 ); // buffer with 2 outputs
notif1 ( z_bar, in, ctrl ); // not with output control
```

برای طراحی مدار در سطح گیت، ابتدا باید آنرا بصورت مجموعه‌ای از گیت‌های پایه درآورد، سپس با تعریف net های مورد نیاز این گیت‌های پایه را به یکدیگر متصل نمود، به این عمل اصطلاحاً Wiring یا سم‌بندی گفته می‌شود. اکنون با ذکر چند مثال نحوه طراحی مدار در سطح گیت را بیان می‌کنیم.

مثال ۱: یک مالتی‌پلکسر ۴ به ۱ طراحی کنید.


حل: طرح شماتیک مدار در سطح گیت و کد Verilog مربوط به آن در شکل ۸-۱ آمده است.

مثال ۲: یک مقایسه‌کننده تک بیتی قابل توسعه طراحی کنید.

حل: طرح شماتیک مدار در سطح گیت و کد Verilog مربوط به آن در شکل ۹-۱ آمده است.

- با استفاده از گیت‌های nand یک فلیپ‌فلاپ نوع D که با لبه بالارونده Clk عمل می‌کند بسازید. ورودی‌های مدار D, Clk و خروجی‌های مدار q, q_bar می‌باشد.

تمرین ۴



```
// Module 4_to_1 multiplexer.
module mux4_to_1 ( out, i0, i1, i2, i3, s1, s0 );
output out;
input i0, i1, i2, i3, s1, s0;

// Internal wire declaration
wire s0n, s1n, y0, y1, y2, y3;


// Gate instantiations
not ( s1n, s1 );
not ( s0n, s0 );

and ( y0, i0, s1n, s0n );
and ( y1, i1, s1n, s0 );
and ( y2, i2, s1, s1n );
and ( y3, i3, s1, s0 );

or ( out, y0, y1, y2, y3 );

endmodule
```

شکل ۸-۱- نقشه شماتیک مالتی پلکسر ۴ به ۱ در سطح گیت و کد Verilog مربوطه



```
//Module 1-bit comparator
module Comparator( a_gt_b, a_eq_b, a_lt_b,
a, b, gt, eq, lt );

output a_gt_b, a_eq_b, a_lt_b;
input a, b, gt, eq, lt;

// Internal wire declaration
wire im0,im1,im2,im3,im4,im5,im6,im7,im8,im9;

// Gate instantiation
not ( im0, b );
not ( im1, a );
nand ( im2, a, gt );
nand ( im3, gt, im0 );
nand ( im4, a, im0 );
nand ( im5, a, b, eq );
nand ( im6, im0, eq, im1 );
nand ( im7, im1, b );
nand ( im8, im1, lt );
nand ( im9, lt, b );
nand ( a_gt_b, im2, im3, im4 );
nand ( a_eq_b, im5, im6 );
nand ( a_lt_b, im7, im8, im9 );

endmodule
```


شکل ۹-۱- نقشه شماتیک مقایسه‌گر تک بیتی قابل توسعه در سطح گیت و کد Verilog مربوطه

۲-۵ تاخیر گیتها

تاکنون گیت‌هایی که در نظر گرفتیم حالت ایده‌آل داشتند، یعنی به محض تغییر ورودی بدون هیچ تاخیری خروجی تغییر پیدا می‌کرد. ولی در مدارهای عملی وضعیت به این‌صورت نیست. تعیین تاخیر برای گیتها اجازه می‌دهد شبیه‌سازی مدارها حالت واقعی‌تری به‌خود بگیرد. مقادیر تاخیر در Verilog با علامت # شروع می‌شود.

**۱-۲-۵ تاخیرهای Rise, Fall, Turn-Off**

- هرگونه تغییر در خروجی گیت از 0, x, z به 1 با تاخیر Rise انجام می‌شود.
- هرگونه تغییر در خروجی گیت از 1, x, z به 0 با تاخیر Fall انجام می‌شود.
- هرگونه تغییر در خروجی گیت به z با تاخیر Turn-Off انجام می‌شود.
- اگر خروجی گیت به X تغییر وضعیت بدهد، مینیموم این سه تاخیر در نظر گرفته می‌شود.
- به هنگام نمونه‌سازی از گیت اگر :
 - فقط یک تاخیر مشخص شود، این مقدار برای تمام تاخیرها در نظر گرفته می‌شود.
 - دو تاخیر مشخص شود، به ترتیب برای تاخیرهای Rise و Fall در نظر گرفته می‌شوند و تاخیر Turn-Off برابر مینیموم این دو مقدار در نظر گرفته می‌شود.
 - سه تاخیر مشخص شود، به ترتیب برای تاخیرهای Rise و Fall و Turn-Off در نظر گرفته می‌شوند.

**۲-۲-۵ مقادیر Min/Typ/Max**

در Verilog هر یک از تاخیرها دارای یک مقدار مینیموم، یک مقدار معمولی و یک مقدار ماکزیموم است. این تاخیرها دارای شکل کلی $Min : Typ : Max$ هستند. انتخاب یکی از این تاخیرها در زمان اجرا صورت می‌گیرد. مثلاً در بعضی شبیه‌سازها وقتی اجرا با سوئیچ `+maxdelay` انجام شود، برای تمام گیتها مقدار ماکزیموم تاخیرها در نظر گرفته می‌شود.

```

// R=Rise Dealy      F=Fall Dealy      T=Turn-Off Delay

// One delay is specified
// if +mindelay, R = F = T = 4
// if +typdelay, R = F = T = 5
// if +maxdelay, R = F = T = 6
and # (4:5:6) ( o1, i1, i2 );

// Two delays are specified
// if +mindelay, R = 3, F = 5, T = min(3,5)
// if +typdelay, R = 4, F = 6, T = min(4,6)
// if +maxdelay, R = 5, F = 7, T = min(5,7)
and # ( 4:5:6, 5:6:7 ) ( o1, i1, i2 );

```




```
// Three delays are specified
// if +mindelay, R = 2, F = 3, T = 4
// if +typdelay, R = 3, F = 4, T = 5
// if +maxdelay, R = 4, F = 5, T = 6
and #( 2:3:4, 3:4:5, 4:5:6 ) ( o1, i1, i2 );
```

- توجه کنید که تاخیر گیتها فقط به هنگام شبیه‌سازی مدارها مفید هستند، و وقتی هدف سنتز مدار می‌باشد به هیچ‌وجه نباید از تاخیر در توصیف مدار استفاده شود.



۶ مدل‌سازی در سطح جریان داده

در مدل‌سازی در سطح جریان داده به نحوه انتقال اطلاعات بین ثباتها و پردازش اطلاعات اهمیت می‌دهیم. در این قسمت به چگونگی مدل‌سازی در سطح جریان داده در زبان Verilog و مزایای آن می‌پردازیم.

۱-۶ Continuous Assignment

توسط این دستور می‌توان یک مقدار را روی یک net درایو کرد. شکل کلی این دستور بصورت

زیر است :

```
assign <signal_strength> <delay> assignment_lists;
```

- در مورد این دستور باید توجه داشت که این دستور همواره فعال است و هرگاه یکی از اپرندهای سمت راست دستور تغییر کند، کل عبارت سمت راست مجددا ارزیابی شده در متغیر سمت چپ قرار می‌گیرد.
- سمت چپ این دستور باید یک متغیر یا یک بردار از نوع net و یا الحاقی از این دو نوع باشد.



```
// A dataflow description of a 2 inputs and gate
module myand( out, i1, i2 );
    output out;
    input i1, i2;
    assign out = i1 & i2;
endmodule
```

۲-۶ Implicit Continuous Assignment

در این روش بجای اینکه یک متغیر را از جنس net تعریف کنیم و سپس توسط دستور assign یک مقدار را روی آن درایو کنیم، می‌توانیم این عمل را در هنگام تعریف متغیر net انجام دهیم.




```
wire out;
assign out = i1 & i2;
// are same as
wire out = i1 & i2;
```

۳-۶ تاخیرها

همانطور که در سطح گیت دیدیم تاخیرها برای این استفاده می‌شوند که عملکرد مدار به واقعیت نزدیکتر باشد. در سطح جریان داده نیز می‌توان برای assignment ها تاخیر مشخص نمود.

۱-۳-۶ تاخیر با قاعده

در اینحالت یک تاخیر را پس از assign و قبل از net می‌آوریم. هر تغییری که در یکی از سیگنالهای سمت راست رخ دهد، باعث می‌شود پس از گذشت زمان تاخیر، عبارت سمت راست مجدداً ارزیابی شود و سپس در متغیر سمت چپ قرار گیرد. به این ترتیب تاخیری که در اینجا داریم، از نوع Inertial است و این امر باعث می‌شود که پالسهایی با عرض کمتر از مقدار تاخیر مشخص شده به خروجی منتشر نشود. این امر در شکل ۱-۱-۱۰ نمایش داده شده است.



شکل ۱-۱-۱۰- تاخیر انتشار Inertial

۲-۳-۶ تاخیر ضمنی

در اینحالت تاخیر و assignment روی یک net به هنگام تعریف آن net مشخص می‌شود.

```
wire out;
assign #5 out = i1 & i2;
// are same as
wire #5 out = i1 & i2;
```

۳-۳-۶ تاخیر به هنگام تعریف net

در اینحالت به هنگام تعریف net تاخیر مورد نظر را برای آن مشخص می‌کنیم. از این پس هر تغییری که روی این net انجام شود، با تاخیر مشخص شده اعمال می‌شود. باید توجه داشت که این امر در مورد مدلسازی در سطح گیت نیز قابل استفاده است.



```
wire #5 out;
assign out = i1 & i2;
```

۴-۶ عبارات، اپراتورها و اپرندها

- یک عبارت از ترکیب اپرندها و اپراتورها ساخته می‌شود.
- یک اپرند می‌تواند یکی از انواع داده‌ای باشد که در بخش ۱-۲ به آن اشاره شد.
- اپراتورها روی اپرندها عملیاتی را انجام می‌دهند تا نتیجه مطلوب بدست آید. لیست اپراتورهای موجود در Verilog در جدول ۱-۵ آمده است.

Operator Type	Operator Symbol	Operation Performed	Number of Operands
Arithmetic	*	multiply	2
	/	division	2
	+	add	2
	-	subtract	2
	%	modulus	2
Logical	!	logical negation	1
	&&	logical and	2
		logical or	2
Relational	>	greater than	2
	<	less than	2
	>=	greater than or equal	2
	<=	less than or equal	2
Equality	==	equality	2
	!=	inequality	2
	===	case equality	2
	!==	case inequality	2
Bitwise	~	bitwise negation	1
	&	bitwise and	2
		bitwise or	2
	^	bitwise xor	2
	~^ or ^~	bitwise xnor	2
Reduction	&	reduction and	1
	~&	reduction nand	1
		reduction or	1
	~	reduction nor	1
	^	reduction xor	1
	~^ or ^~	reduction xnor	1
Shift	>>	right shift	2
	<<	left shift	2
Reduction	{ }	concatenation	any number
	{ { } }	replication	any number
Conditional	? :	conditional	3

جدول ۱-۵- لیست اپراتورهای Verilog برای مدلسازی در سطح جریان داده

**نکته**

- در مورد اپراتورهای حسابی باید به موارد زیر توجه داشت :
- اگر هردو اپرندش صحیح باشد، خارج قسمت را برمی گرداند.
- اگر هر یک از اپرندها دارای بیت X باشد، نتیجه عملیات X خواهد بود.
- % باقیمانده تقسیم را برمی گرداند و علامت حاصل برابر علامت اپرند اول است.
- بهتر است اعداد منفی را در عبارات بصورت اعداد صحیح بکار برد، زیرا در غیراینصورت به مکمل ۲ تبدیل می شوند که ممکن است باعث بروز نتایج غیرمنتظره شوند.

```
A = 4'b0011;   B = 4'b0100;   D = 6;   E = 4;
A*B           // Evaluated to 4'b1100
D/E           // Evaluated to 1
A+B           // Evaluated to 4'b0111
B-A           // Evaluated to 4'b0001
//-----
in1 = 4'b101x;  in2 = 4'b1010;
sum = in1 + in2; // Evaluated to 4'x
//-----
-7 % 2 // Evaluated to -1, take sign of the first operand
7 % -2 // Evaluated to +1, take sign of the first operand
```

نکته

- در مورد اپراتورهای منطقی باید به موارد زیر توجه داشت :
- نتیجه اپراتورهای منطقی یک بیت است : 0 نادرست، 1 درست، X نامعلوم.
- اگر اپرند 0 باشد معادل نادرست، اگر 1 باشد معادل درست و اگر X باشد معادل نامعلوم ارزیابی می شود.

```
A = 3;   B = 0;
A && B // Evaluated to 0 ( False )
A || B // Evaluated to 1 ( True )
!A     // Evaluated to 0 ( False )
//-----
A = 2'bx0;   B = 2'b10;
A && B // Evaluated to x ( Unknown )
```

نکته

- در مورد اپراتورالحاق باید به موارد زیر توجه داشت :
- اپرندها حتما باید اعداد اندازه دار باشند تا Verilog قادر به محاسبه اندازه نتیجه باشد.
- اپرندها می توانند net ، reg ، برداری از net ، reg یا اعداد اندازه دار باشند.



```
A = 1'b1;   B = 2'b00;   D = 2'b10;   C = 3'b110;
y = { B, C };                               // Result y is 4'b0010
y = { A, B ,C, D, 3'b001 };                 // Result y is 11'b1_00_10_110_001
y = { A, b[0], C[1] };                     // Result y is 3'b101
```

- در مورد اپراتور تکرار باید به موارد زیر توجه داشت :
- یک ثابت تکرار مشخص می‌کند که چندبار عدد داخل {} باید تکرار شود.



```
A = 1'b1;   B = 2'b00;   D = 2'b10;

y = { 4{A} };                               // Result y is 4'b1111
y = { 4{A}, 2{B} };                         // Result y is 8'b11110000
y = { 4{A}, 2{B}, C };                     // Result y is 10'b1111000010
```

- در مورد اپراتور شرطی باید به موارد زیر توجه داشت :
- قالب استفاده از اپراتور شرطی بصورت زیر است :
`<condition> ? true_exp : false_exp;`
- عبارت `condition` ارزیابی می‌شود، اگر نتیجه عبارت `x` باشد، هر دو عبارت `true_exp`, `false_exp` محاسبه شده و بیت به بیت با هم مقایسه می‌شوند، اگر بیتها باهم متفاوت بودند `x` و در غیر اینصورت همان بیت برگردانده می‌شود.



```
// Models functionality of a tri state buffer
assign addr_bus = drive_enable ? addr_out : 32'bz;
```

- در مورد اپراتورهای تساوی باید به موارد زیر توجه داشت :
- نتیجه `a==b` برابر 0 یا 1 یا `x` است. اگر یک بیت از یکی از ابرندها `x` یا `z` باشد، نتیجه `x` می‌شود.
- نتیجه `a!=b` برابر 0 یا 1 یا `x` است. اگر یک بیت از یکی از ابرندها `x` یا `z` باشد، نتیجه `x` می‌شود.
- نتیجه `a===b` برابر 0 یا 1 است. `a` و `b` بیت به بیت با هم مقایسه می‌شوند.
- نتیجه `a!===b` برابر 0 یا 1 است. `a` و `b` بیت به بیت با هم مقایسه می‌شوند.





```
A = 3;    B = 3;
X = 4'b1010;  Y = 4'b1101;
Z = 4'b1xzz;  M = 4'b1xzz;    N = 4'b1xxx;

A == B      // Result is logical 0
A != B      // Result is logical 1
X == Z      // Result is x
Z === M     // Result is logical 1
Z === N     // Result is logical 0
M !== M     // Result is logical 1
```

- در مورد اپراتورهای کاهش (reduction) باید به موارد زیر توجه داشت :
- این اپراتورها دارای یک اپرند هستند و عملیات بیتی مشخص شده را روی تک تک اعضای بردار اپرند آن انجام داده و یک بیت را بعنوان نتیجه برمی گردانند.



```
x = 4'b1010;
&x      // Equivalent to 1 & 0 & 1 & 0, Result is 1'b0
|x      // Equivalent to 1 | 0 | 1 | 0, Result is 1'b0
^x      // Equivalent to 1 ^ 0 ^ 1 ^ 0, Result is 1'b0
```

حال سعی می‌کنیم با ذکر چند مثال به نحوه مدلسازی مدارهای دیجیتال در سطح جریان داده

بپردازیم.

مثال ۱: یک مالتی پلکسر ۴ به ۱ طراحی کنید.

```
// Dataflow model of a 4-to-1 multiplexer

module mux4_to_1 ( out, i0, i1, i2, i3, s1, s0 );
    output out;
    input i0, i1, i2, i3, s1, s0;

    // Use nested conditional operator
    assign out = s1 ? ( s0 ? i3 : i2 ) : ( s0 ? i1 : i0 );

endmodule
```

مثال ۲: یک جمع‌کننده ۴ بیتی طراحی کنید.



```
// Dataflow model of a 4-bit full adder

module fulladd4 ( sum, c_out, a, b, c_in );
    output [3:0] sum;
    output c_out;
    input [3:0] a, b;
    input c_in;

    // Specify the function of a 4-bit full adder
    assign { c_out, sum } = a + b + c_in;

endmodule
```

مثال ۳: یک مقایسه‌کننده تک بیتی قابل توسعه طراحی کنید.


```
// Dataflow model of a cascadable 1-bit comparator

module Comparator ( a_gt_b, a_eq_b, a_lt_b, a, b, gt, eq, lt );
    output a_gt_b, a_eq_b, a_lt_b;
    input a, b, gt, eq, lt;

    // Specify the boolean function of a 1-bit comparator
    assign a_gt_b = ( a & gt ) | ( ~b & gt ) | ( a & ~b );
    assign a_eq_b = ( a & b & eq ) | ( ~a & ~b & eq );
    assign a_lt_b = ( ~a & lt ) | ( b & lt ) | ( ~a & b );

endmodule
```

- در شکل زیر طرح شماتیک مداری که مکمل ۲ یک عدد باینری ۴ بیتی را محاسبه می‌کند نشان داده شده است. این مدار را در سطح جریان داده توصیف کنید.



پرسش ۵



۷ مدل‌سازی در سطح رفتاری

۱-۷ بلوکهای ساخت یافته

در Verilog دو بلوک ساخت یافته وجود دارد، `initial`، `always`. این دستورات پایه مدل‌سازی در سطح رفتاری هستند و تمام قسمت‌های توصیف رفتاری مدار در این بلوکهای ساخت یافته قرار می‌گیرند. این بلوکها دارای ساختار ترتیبی هستند.

۱-۱-۷ بلوک `initial`

بلوک `initial` در زمان 0 شروع شده و فقط یکبار در شروع شبیه‌سازی اجرا می‌شود. چنانچه بخواهیم درون بلوک `initial` چندین دستور داشته باشیم، باید بوسیله `begin end` آنها را بلوک کنیم. چنانچه چندین بلوک `initial` داشته باشیم، تمام بلوکها در زمان 0 بصورت همزمان اجرا می‌شوند و هر بلوک مستقل از سایر بلوکها خاتمه می‌یابد. اگر قبل از یک دستور تاخیری مشخص شود، آن دستور پس از تاخیر مشخص شده از زمان فعلی شبیه‌سازی اجرا می‌شود.

```
module stimulus;

    reg a, b, x, y, m;

    initial
        m = 1'b0;

    initial
    begin
        #10 x = 1'b0;
        #10 y = 1'b1;
    end

    initial
    begin
        #5 a = 1'b1;
        #10 b = 1'b0;
    end

    initial
        #50 $finish;

endmodule

//      Time      Statement executed
//      0          m = 1'b0;
//      5          a = 1'b1;
//      10         x = 1'b0;
//      30         b = 1'b0;
//      35         y = 1'b1;
//      50         $finish;
```




۲-۱-۷ بلوک always

بلوک always در زمان 0 شروع شده و تمام دستورات درون آن بطور پیوسته اجرا می‌شوند (مانند یک حلقه). این دستورات برای مدل‌سازی یک بلوک از اعمال که متوالیا در یک مدار انجام می‌شوند، بکار می‌رود.

```
module clock_gen;

    reg clk;

    initial
        clk = 1'b0;

    // Toggle clk every half_cycle ( Period = 20 )
    always
        #10 clk = ~clk;

    initial
        #1000 $finish;

endmodule
```

۲-۷ Procedural Assignment

این دستور مقدار یک متغیر reg، صحیح، حقیقی یا زمان را تغییر می‌دهد. مقدار جدید در متغیر باقی می‌ماند تا هنگامیکه یک دستور دیگر مقدار آن را تغییر دهد. سمت چپ این عبارت می‌تواند یکی از موارد زیر باشد:

- یک متغیر reg، صحیح، حقیقی، زمان یا عناصر حافظه
- یک Bit-Select از این متغیرها
- یک Part-Select از این متغیرها
- الحاقی از موارد فوق

۱-۲-۷ Blocking Assignment

این دستورات به همان ترتیبی که مشخص شده‌اند اجرا می‌شوند، یعنی بدون کامل شدن اجرای یک دستور، دستور بعدی اجرا نمی‌شود. اپراتوری که برای این امر بکار می‌رود = است.

```
initial
begin

    // These statements are executed at time 0 sequentially
    x = 0;    y = 1;    z = 1;
    count = 0;
    reg_a = 16'h0000;  reg_b = reg_a;

    // This statement is executed at time 15
    #15 reg_a[2] = 1'b1;
```



```
// These statements are executed at time 25 sequentially
#10 reg_b[15:13] = { x, y, z };
count = count + 1;

end
```

۲-۲-۷ Nonblocking Assignment

در این روش کلیه دستورات برای اجرا زمانبندی می‌شوند، بدون اینکه منتظر کامل شدن اجرای یک دستور باشیم. اپراتوری که برای این امر بکار می‌رود \leq است.

- شبیه‌ساز یک دستور Nonblocking را برای اجرا زمانبندی می‌کند، سپس به دستور بعدی درون بلوک می‌پردازد، بدون اینکه منتظر کامل شدن اجرای دستور قبلی شود.
- دستورات Nonblocking می‌توانند بصورت مؤثری همزمانی انتقال اطلاعات را مدل کنند، زیرا نتیجه نهایی به ترتیب اجرای دستورات وابسته نیست.



```
initial
begin

    x = 0;    y = 1;    z = 1;
    count = 0;
    reg_a = 16'h0000;  reg_b = reg_a;

    // This statement is scheduled to execute after 15 time units
    reg_a[2] <= #15 1'b1;

    // This statement is scheduled to execute after 10 time units
    reg_b[15:13] <= #10 { x, y, z };

    // This statement is scheduled to execute without any delay
    count <= count + 1;

end
```

۳-۷ کنترل زمان

در Verilog چنانچه دستورات کنترل زمان موجود نباشد، شبیه‌سازی انجام نمی‌شود. در Verilog سه نوع کنترل زمان وجود دارد:

۱-۳-۷ کنترل زمان مبتنی بر تاخیر

در این روش یک عبارت، فاصله زمانی بین رسیدن به یک دستور تا اجرای آن را مشخص می‌کند. دو نوع کنترل زمان مبتنی بر تاخیر موجود است.



- **کنترل تاخیر با قاعده:** یک تاخیر غیر صفر در سمت چپ دستور آورده می‌شود. در اینحالت عبارت سمت راست پس از گذشت زمان تاخیر، محاسبه شده درون عبارت سمت چپ قرار می‌گیرد.

```
initial
begin
  x = 0;
  #10 y = 1;
  #(4:5:6) q=0;
end
```

- **کنترل تاخیر درون دستور:** یک تاخیر غیر صفر در سمت راست اپراتور assignment آورده می‌شود. در اینحالت عبارت سمت راست در زمان فعلی محاسبه شده، پس از گذشت زمان تاخیر درون عبارت سمت چپ قرار می‌گیرد.

```
initial
begin
  x = 0;   z = 0;
  y = #5 x + z;
end
// Is equivalent to
initial
begin
  x = 0;   z = 0;   temp = x + z;
  #5 y = temp;
end
```

۲-۳-۷ کنترل زمان مبتنی بر رویداد

یک رویداد به معنای تغییر مقدار یک reg یا net است. چهارنوع کنترل زمان مبتنی بر رویداد وجود دارد.

- **کنترل رویداد با قاعده:** علامت @ برای مشخص کردن کنترل رویداد استفاده می‌شود. دستورات می‌توانند با تغییر مقدار یک سیگنال، با لبه بالارونده یا پایین رونده یک سیگنال اجرا شوند. لبه بالارونده به معنی یکی از تغییرات $0 \rightarrow 1, x, z; x \rightarrow 1; z \rightarrow 1$ و لبه بالارونده به معنی یکی از تغییرات $1 \rightarrow 0, x, z; x \rightarrow 0; z \rightarrow 0$ می‌باشد.

```
@(clock) q = d;           // Triggered with any change in clock
@(posedge clock) q = d;  // Triggered positive edge of clock
@(negedge clock) q = d;  // Triggered negative edge of clock
q = @(posedge clock) d;   // d is evaluated immediately and
                          // assigned to q at negative edge of clock
```

- **کنترل رویداد با نام:** Verilog این امکان را برای ما فراهم ساخته است که یک رویداد را تعریف کنیم و در موقع لزوم آنرا تریگر کنیم. تعریف رویداد با کلمه کلیدی event و تریگر کردن آن با >- انجام می‌شود.



```
event rec_data;

always @(posedge clock)
begin
    if( last_data_packet ) ->rec_data;
end

always @(rec_data)
data_buf = { data[0] , data[1], data[2], data[3] };
```

- **کنترل چند رویداد**: گاهی اوقات چند سیگنال داریم که تغییر در یکی از آنها سبب تریگر شدن اجرای یک مجموعه از دستورات می‌شود. این امر توسط **or** کردن رویدادها یا سیگنالها انجام می‌شود. لیست رویدادها یا سیگنالها به **Sensitivity List** مشهور است.

```
always @(posedge clock or reset)
begin
    if( reset ) q = 0;
    else q = d;
end
```

۳-۳-۷ کنترل حساس به سطح

- Verilog دارای این قابلیت است که اجرای یک دستور را تا تحقق یک شرط خاص به تعویق بیندازیم، این امر توسط دستور **wait** انجام می‌شود.

```
always
    wait( count_enable ) #20 count = count + 1;
```

۴-۷ دستور شرطی

```
if( expr )
    true_st
else
    false_st;
```

۵-۷ دستور case

```
case ( expr )
    case 1 : st1;
    case 2 : st2;
    ...
    case n : stn;
    default : def_st;
endcase
```



- عبارت `expr` با تک تک موارد بیت به بیت مقایسه می‌شود (با در نظر گرفتن `0, 1, x`).
- اگر طول عبارت با موارد مورد مقایسه یکسان نباشد، سمت چپ جزء کوچکتر با `0` پر می‌شود.
- در صورت استفاده از `case x` مقادیر `x, z` بصورت بی‌اهمیت (`d'ont care`) تلقی می‌شوند یعنی در مقایسه نقشی ندارند.
- در صورت استفاده از `case z` مقدار `z` بصورت بی‌اهمیت (`d'ont care`) تلقی می‌شود یعنی در مقایسه نقشی ندارد.

۶-۷ حلقه‌ها

```
while( expr )
    st;

//-----
for( init; end_cond; chang_control_var )
    st;

//-----
repeat( number_of_iteration )
    st;

//-----
forever
    st;
```

۷-۷ Tasks and Functions

Task و تابع دارای تفاوت‌هایی هستند که در جدول ۶-۱ نشان داده شده است.

Task ۱-۷-۷

- بوسیله `task ... endtask` مشخص می‌شود. معمولاً در موارد زیر استفاده می‌شود.
- در پروسیجر تاخیر، زمانبندی و یا کنترل زمان وجود داشته باشد.
 - پروسیجر پارامتر خروجی نداشته باشد و یا بیش از یک پارامتر خروجی داشته باشد.
 - پروسیجر هیچ پارامتر ورودی نداشته باشد.



Function	Task
یک تابع فقط می‌تواند توابع دیگر را فعال کند.	یک task می‌تواند توابع و task های دیگر را فعال کند.
یک تابع همیشه در زمان 0 اجرا می‌شود.	یک task ممکن است در زمان 0 اجرا می‌شود.
یک تابع نمی‌تواند تاخیر، رویداد و یا کنترل زمان داشته باشد.	یک task می‌تواند تاخیر، رویداد و یا کنترل زمان داشته باشد.
یک تابع حداقل باید یک آرگومان ورودی داشته باشد.	یک task می‌تواند آرگومانهای ورودی، خروجی یا دوسویه داشته باشد.
یک تابع همیشه مقدار بازگشتی دارد.	یک task نمی‌تواند مقدار بازگشتی داشته باشد.

جدول ۶-۱- تفاوت‌های Task و تابع

```

module operation;
  reg [15:0] A, B;
  reg [15:0] AB_AND, AB_OR, AB_XOR;

  always @(A or B) //whenever A or B changes in value
    bitwise_oper(AB_AND, AB_OR, AB_XOR, A, B);

  // Define task bitwise_oper
  task bitwise_oper;
    output [15:0] ab_and, ab_or, ab_xor; //outputs from the task
    input [15:0] a, b; //inputs to the task
  begin
    #20 ab_and = a & b;
        ab_or = a | b;
        ab_xor = a ^ b;
  end
  endtask

endmodule

```

Function ۲-۷-۷

بوسیله `function ... endfunction` مشخص می‌شود. معمولاً در موارد زیر استفاده می‌شود.

- در پروسیجر تاخیر، زمانبندی و یا کنترل زمان وجود نداشته باشد.
- پروسیجر یک پارامتر خروجی داشته باشد.
- پروسیجر حداقل یک پارامتر ورودی داشته باشد.

```

function calc_parity;
input [31:0] address;
begin

  // set the output value appropriately. Use the implicit
  // internal register calc_parity.

  calc_parity = ^address; //Return the ex-or of all address bits.

end
endfunction

```



۸-۷ طراحی مدارهای ترکیبی در سطح رفتاری

برای طراحی مدارهای ترکیبی در سطح رفتاری، باید تمام ورودی‌های مدار را در لیست حساس بدنه‌ی `always` ذکر کرد. به هنگام توصیف مدار باید توجه داشت که تمام شرط‌های `if` باید دارای `else` باشند تا از ایجاد مدار ترتیبی جلوگیری شود.
مثال: یک مالتی‌پلکسر ۴ به ۱ طراحی کنید.

```
// Behavioral model of a 4-to-1 multiplexer
module mux4_to_1 ( out, i0, i1, i2, i3, s1, s0 );
    output out;
    reg out;
    input i0, i1, i2, i3, s1, s0;

    always @( i0 or i1 or i2 or i3 or s1 or s0 )
    begin
        case { s1, s0 }
            2'b00 : out = i0;
            2'b01 : out = i1;
            2'b10 : out = i2;
            2'b11 : out = i3;
            default : out = 1'bx;
        endcase
    end
endmodule
```

۹-۷ طراحی مدارهای ترتیبی در سطح رفتاری

طراحی مدارهای ترتیبی را با استفاده از چند مثال بیان می‌کنیم.
مثال ۱: یک D-FF حساس به لبه‌ی بالارونده با `reset` سنکرون طراحی کنید.

```
// Behavioral model of a d-ff with synchronous reset
module d_ff ( d, clk, rst, q );
    input d, clk, rst;
    output q;
    reg q;

    always @( posedge clk )
    begin
        if (rst)
            q = 1'b0;
        else
            q = d;
    end
endmodule
```

مثال ۲: یک D-FF حساس به لبه‌ی بالارونده با `reset` آسنکرون طراحی کنید.



```
// Behavioral model of a d-ff with asynchronous reset

module d_ff ( d, clk, rst, q );
  input d, clk, rst;
  output q;
  reg q;

  always @( posedge clk or posedge rst)
  begin
    if (rst)
      q = 1'b0;
    else
      q = d;
  end
endmodule
```


مثال ۳: یک شمارنده‌ی ۴ بیتی با پایه‌های سنکرون `ld`, `u_d` و پایه‌ی `rst` آسنکرون طراحی کنید که با لبه‌ی پایین‌رونده‌ی `clk` کار کند.

```
// Behavioral model of a up/down counter

module counter ( clk, ld, rst, u_d, d_in, q );
  input clk, ld, rst, u_d;
  input [3:0] d_in;
  output [3:0] q;
  reg [3:0] q;

  always @( negedge clk or posedge rst)
  begin
    if (rst)
      q = 4'b0000;
    else if( ld )
      q = d_in;
    else if( u_d )
      q = q + 1;
    else
      q = q - 1;
  end
endmodule
```

مثال ۴ (طراحی دیاگرام حالت mealy): دیاگرام حالت زیر را مدلسازی کنید.





```
// Mealy state machine

`define init          2'd0
`define got1          2'd1
`define got11         2'd2

module seq_detector ( clk, x, rst, y );
    input clk, x, rst;
    output y;

    reg [1:0] cur_state;

    always @( posedge clk )
    begin
        if ( rst )
            cur_state = `init;
        else case ( cur_state )
            `init : cur_state = x ? `got1 : `init;
            `got1 : cur_state = x ? `got11 : `init;
            `got11 : cur_state = x ? `got11 : `init;
        endcase
    end

    assign y = (cur_state==`got11 && x==1'b0) ? 1'b1 : 1'b0;

endmodule
```

مثال ۵: دیاگرام حالت مثال ۴ را با استفاده از مدل هافمن مدلسازی کنید.

```
// Mealy state machine (Huffman Model)

`define init          2'd0
`define got1          2'd1
`define got11         2'd2

module seq_detector ( clk, x, rst, y );
    input clk, x, rst;
    output y;

    reg [1:0] present_state, next_state;

    always @( posedge clk )
    if(rst)
        present_state = `init;
    else
        present_state = next_state;


    always @( present_state or x )
    begin
        case ( present_state )
            `init : next_state = x ? `got1 : `init;
            `got1 : next_state = x ? `got11 : `init;
            `got11 : next_state = x ? `got11 : `init;
        endcase
    end

    assign y = (present_state==`got11 && x==1'b0) ? 1'b1 : 1'b0;

endmodule
```



مثال ۶ (طراحی دیاگرام حالت moore): دیاگرام حالت زیر را مدلسازی کنید.



```
// Moore state machine

`define init      2'd0
`define got1     2'd1
`define got11    2'd2
`define got110   2'd3

module seq_detector ( clk, x, rst, y );
    input clk, x, rst;
    output y;

    reg [1:0] cur_state;

    always @( posedge clk )
    begin
        if ( rst )
            cur_state = `init;
        else case ( cur_state )
            `init    : cur_state = x ? `got1 : `init;
            `got1    : cur_state = x ? `got11 : `init;
            `got11   : cur_state = x ? `got11 : `got110;
            `got110  : cur_state = x ? `got1 : `init;
        endcase
    end

    assign y = (cur_state==`got110) ? 1'b1 : 1'b0;

endmodule
```

مثال ۷: دیاگرام حالت مثال ۶ را با استفاده از مدل هافمن مدلسازی کنید.

```
// Moore state machine (Huffman Model)

`define init      2'd0
`define got1     2'd1
`define got11    2'd2
`define got110   2'd3

module seq_detector ( clk, x, rst, y );
    input clk, x, rst;
    output y;

    reg [1:0] present_state, next_state;

    always @( posedge clk )
        if(rst)
            present_state = `init;
        else
            present_state = next_state;
```



```
always @( present_state or x )
begin
    case ( present_state )
        `init      : next_state = x ? `got1  : `init;
        `got1      : next_state = x ? `got11 : `init;
        `got11     : next_state = x ? `got11 : `got110;
        `got110    : next_state = x ? `got1  : `init;
    endcase
end

assign y = (present_state==`got110) ? 1'b1 : 1'b0;

endmodule
```

مثال ۸: فرض کنید در یک سیستمی یک پروسس تولید کننده و یک پروسس مصرف کننده وجود دارد. می‌خواهیم Handshake بین این دو پروسس را به نوعی مدلسازی کنیم. سیستم به اینصورت عمل می‌کند که ابتدا پروسس تولیدکننده داده‌ای را تولید می‌کند، پروسس مصرف کننده این داده را مصرف می‌کند. مادامیکه پروسس مصرف کننده داده را مصرف نکرده، پروسس تولیدکننده داده بعدی را تولید نمی‌کند.

```
module producer( dataOut, prodReady, consReady )
    output [7:0] dataOut;
    output prodReady;
    reg [7:0] dataOut;
    reg prodReady;
    input consReady;

    always
    begin
        prodReady = 0;          // Indicate nothing to transfer
        forever
        begin
            // ... Produce data and put into temp
            // Wait for consumer ready
            wait( consReady ) dataOut = $random;

            // Indicate ready to transfer
            prodReady = 1;
            // Finish handshake
            wait( !consReady ) prodReady = 0;
        end
    end

endmodule

module consumer( dataIn, prodReady, consReady )
    input [7:0] dataIn;
    input prodReady;
    output consReady;
    reg consReady;

    reg [7:0] dataInCopy;
```



```
always
begin
  consReady = 1;          // Indicate nothing to transfer
  forever
  begin
    wait( prodReady ) dataInCopy = dataIn;
    // Indicate value consumed
    consReady = 0;
    // ... munch on data
    // Complete handshake
    wait( !prodReady ) consReady = 1;
  end
end


endmodule

// Top-level Producer-Consumer process
module ProducerConsumer;
  wire [7:0] data;
  wire pReady, cReady;

  producer p( data, pReady, cReady );
  consumer c( data, pReady, cReady );

endmodule
```

- مثال تولیدکننده و مصرف‌کننده فوق را به نحوی تغییر دهید که در لبه بالا رونده clk های متوالی تولیدکننده و مصرف‌کننده به ترتیب داده را تولید و مصرف کنند (شکل زیر را ببینید).




برسختی ۵

۸ مدل‌سازی در سطح سوئیچ

در Verilog برای انجام مدل‌سازی در سطح سوئیچ ساختارهایی از پیش تعبیه شده است، که در این قسمت به آنها می‌پردازیم.

۱-۸ سوئیچهای MOS, CMOS

در شکل ۱-۱۱ این سوئیچها و جدول صحت آنها نمایش داده شده است.




شکل ۱۱-۱- سوئیچهای MOS و جدول صحت آنها

```
nmos n1 ( out, data, control );
pmos ( out, data, control );
cmos c1 ( out, data, ncontrol, pcontrol );
```

۲-۸ سوئیچهای دوسویه

در شکل ۱۲-۱ سوئیچهای دوسویه نمایش داده شده است.



شکل ۱۱-۱- سوئیچهای MOS و جدول صحت آنها

```
tran t1 ( inout1, inout2 );
tranif0 ( inout1, inout2, control );
tranif1 ( inout1, inout2, control );
```

۳-۸ تغذیه و زمین

به هنگام استفاده از سوئیچها به منبع تغذیه و زمین (Vss, Vdd) نیاز داریم. برای این منظور دو کلمه کلیدی supply0, supply1 به ترتیب برای Vss, Vdd در نظر گرفته شده است.

```
supply1 Vdd;
supply0 Vss;
assign a = Vdd; // Connect a to Vdd
```

**۴-۸ سوئیچهای مقاومتی**

درحالات قبل سوئیچها ایده‌آل فرض شده بودند، یعنی هیچگونه تضعیفی در سیگنال عبوری از سوئیچ مشاهده نمی‌شد. ولی در سوئیچهای مقاومتی قدرت سیگنال عبوری از سوئیچ مطابق جدول ۷-۱ تغییر می‌کند. سوئیچهای مقاومتی با افزودن یک T به ابتدای نام سوئیچهای معمولی بدست می‌آیند.

```

rnmos, rpmos // Resistive nmos,pmos switches
rcmos // Resistive cmos switch
rtran, rtranif0, rtranif1 // Resistive tran,tranif0,tranif1 switches

```

Input Strength	Output Strength
supply	pull
strong	pull
pull	weak
weak	medium
large	medium
medium	small
small	small
high	high

جدول ۷-۱- جدول تضعیف سیگنال سوئیچهای مقاومتی

مثال: یک مالتی پلکسر ۲ به ۱ طراحی کنید.

